
SAS syntax and output for Chapter 6: Introduction to Multilevel Modeling
Much of the material in this document builds on the SAS sytax and output from earlier chapters. If necessary, refer back to those documents for reminders.


Begin by importing the ‘beaut.dat’ file (for the course-evaluation example) into SAS:

	data ch6data;
 infile 'c:\beaut.dat' FIRSTOBS=2 dlm=' ';
 input id $ profnumber eval beauty female age minority tenured gpa;
 drop id;
run;





Descriptive statistics for course evaluation outcome variable (‘eval’) and student GPA (‘gpa’), for now ignoring clustering within instructors:

	proc means data=ch6data maxdec=2 n mean std median min max range skew kurt;
 var eval;
run;






	proc means data=ch6data maxdec=2 n mean std median min max range skew kurt;
 var gpa;
run;






Descriptive statistics on instructor-by-instructor basis:

First, to get stats just for instructors 5, 11, 30, 68, and 85, as presented in Chapter 6, it is convenient to subset the data (‘profnumber’ is the variable giving Instructor ID):


	data profSubset;
 set ch6data;
 where profnumber=5 | profnumber=11 | profnumber=30 | profnumber=68 | profnumber=85;
run;



Then get within-instructor descriptive stats for the course evaluation outcome and the GPA predictor using this subset:

	proc means data=profSubset maxdec=2 n mean std median min max range skew kurt;
 var eval;
 class profnumber;
run;






	proc means data=profSubset maxdec=2 n mean std median min max range skew kurt;
 var gpa;
 class profnumber;
run;






Next, a separate data file (‘ch6j94.txt’) is available which has only one row per instructor and contains only the instructor-level variables. Using this dataset, it is easy to get descriptive stats for the instructor attractiveness variable, here named ‘beauty’ (because Hamermesh and Parker, 2005, named it beauty):


	data ch6j94;
 infile ''c:\beaut-j94.dat' firstobs=2 dlm=' ';
 input id $ profnumber eval beauty female age minority tenured gpa;
 drop id;
run;

proc means data=ch6j94 maxdec=2 n mean std median min max range skew kurt;
 var beauty;
run;






But because course evaluation is a Level 1 variable, we need to return to the full dataset to get the correlation between instructor attractiveness and course evaluation and make the scatterplot in Figure 6.1:

	data ch6data;
 set ch6data;
 evaln = eval;
run;

proc sgplot data=ch6data;
 scatter x=evaln y=beauty; 
 xaxis label="Instructor physical attractiveness";
 yaxis label="Course evaluation";
run;
[image: ]





	proc corr data=ch6data;
 var eval beauty;
run; 






(The significance test of the correlation above is biased because the independent observations assumption is violated.)

Scatterplot of course evaluation by student GPA, ignoring clustering of students within instructors (Figure 6.2):

	proc sgplot data=ch6data;
 scatter x=gpa y=evaln; 
 xaxis label="Student GPA";
 yaxis label="Course evaluation";
run;
[image: ]



Plot of course evaluation by student GPA, within each of four instructors (Figure 6.3).
First, make a subset of the data, selecting only the instructors whose data will be plotted:

	data prof2050;
 set ch6data;
 where profnumber=20 | profnumber=50 | profnumber=34 | profnumber=82;
run;



The graph in Figure 6.3 can be made with the ‘sgpanel’ procedure:

	proc sgpanel data = prof2050;
  panelby profnumber;
  scatter x=gpa y=evaln; 
  reg x=gpa y=evaln; 
  rowaxis label="Course evaluation";
  colaxis label="Student GPA";
run;
[image: ]





The Unconditional Multilevel Model

After examining descriptive statistics and plots, we are ready to begin estimating multilevel models. 

The multilevel models in this chapter can all be estimated using the ‘mixed’ procedure:

	proc mixed data=ch6data; 
 class profnumber;
 model eval = / solution;
 random intercept / subject=profnumber; 
run;



Above, the ‘class’ line is used so that ‘profnumber’ is treated as a categorical variable. Because this is an intercept-only model, the outcome variable (‘eval’) is regressed on (=) only a constant value that is not written the syntax. Next, the statement  ‘random intercept / subject=profnumber’ indicates that the model should allow intercepts to vary randomly across the levels of instructor ID (i.e., ‘profnumber’, the Level 2 ID variable). By default, the model is estimated using REML. See below for an example where the estimator is changed to FIML.

The ‘proc mixed’ output is below:

	




In the output above, the only fixed effect estimate is the mean intercept value, i.e., gamma, the grand mean of the course evaluation outcome.

Under ‘Covariance Parameter Estimates’ (random effects), the Intercept estimate corresponds to the Level 2 variance estimate in Table 6.2, while the Residual estimate corresponds to the Level 1 variance estimate.

And we can use these values from the ‘Covariance Parameter Estimates’ to calculate the ICC:

	data temp;
 icc = 0.1470 / (0.1470+0.1702);
run;

proc print data=temp;
run;





The resulting value matches the ICC value given in Chapter 6.



Conditional Multilevel Models

Specify and estimate the random-intercepts MLM regressing course evaluation data on student GPA.

The model specification below is a simple elaboration on the unconditional model, now including student GPA as a predictor:

	proc mixed data=ch6data; 
 class profnumber;
 model eval = gpa / solution;
 random intercept / subject=profnumber; 
run;





The fixed effects results above give the mean intercept and GPA coefficient reported in Table 6.3, while again the variance estimates of the random intercepts (the Level 2 variance estimate) and the Level 1 variability are given under the ‘Covariance Parameter Estimates’ output.

Confidence intervals can be obtained by adding the ‘cl’ option on the ‘model’ line:

	proc mixed data=ch6data; 
 class profnumber;
 model eval = gpa / solution cl;
 random intercept / subject=profnumber; 
run;






Above, ‘Lower’ and ‘Upper’ give the 95% confidence limits for the corresponding effects.


Specify and estimate the random-slopes MLM regressing course evaluation data on student GPA.

The model specification below is a simple elaboration on the random-intercepts model, now including GPA as a random effect:

	proc mixed data=ch6data; 
 class profnumber;
 model eval = gpa / solution cl;
 random intercept gpa / subject=profnumber; 
run;



As reported in Chapter 6, the model specified above does not converge using the default REML estimator with the default number of iterations.

For the sake of demonstration, we can also try the FIML estimator by using the ‘method=ML’ argument:

	proc mixed data=ch6data method=ML; 
 class profnumber;
 model eval = gpa / solution cl;
 random intercept gpa / subject=profnumber; 
run;



But the model still does not converge.
It turns out, though, that this model does converge if we increase the number of iterations. Later we will see how to increase the number of iterations.



Model with Level 2 predictor (means-as-outcomes model)

Specify and estimate model regressing course evaluation on instructor attractiveness (‘beauty’):

	proc mixed data=ch6data; 
 class profnumber;
 model eval = beauty / solution cl;
 random intercept / subject=profnumber; 
run;






The output above provides the results in Table 6.4 and the 95% CI reported in the text for the attractiveness fixed effect.



Model with both a Level 1 predictor and a Level 2 predictor

Specify and estimate the model regressing course evaluation on both student GPA and instructor attractiveness (‘beauty’). Here, the model has random intercepts only; at this point, random-slopes for the GPA effect make the model too complex:

	proc mixed data=ch6data; 
 class profnumber;
 model eval = gpa beauty / solution cl;
 random intercept / subject=profnumber; 
run;








The output above provides the results in Table 6.5 and the 95% CIs reported in the text.



Distinguishing Within-Cluster Effect from Between-Clusters Effect

The syntax below can be used to cluster-mean center a Level 1 variable and create the Level 2 cluster means from a Level 1 variable:

	data ch6data;
 set ch6data;
 id=_n_;
run;

proc sort data=ch6data;
 by profnumber;
run;

proc means data=ch6data mean;
 var gpa;
 by profnumber;
 output out=grpmean mean=m1;
run;

proc sort data=ch6data;
 by profnumber id;
run;

data ch6data;
 merge ch6data grpmean;
 by profnumber;
 drop _TYPE_ _FREQ_;
 gpawc = gpa - m1;
 gpameans = m1;
run;



The final data step creates a new variable, called ‘gpawc’, which is the cluster-mean centered version of the original ‘gpa’ variable and a new Level 2 variable, called ‘gpameans’, which is the mean GPA for each instructor.

Now we can use the variables created above to specify and estimate the random-intercepts model for separate within-cluster, Level 1 GPA effect and between-clusters, Level 2 GPA effect:

	proc mixed data=ch6data; 
 class profnumber;
 model eval = gpawc gpameans / solution cl;
 random intercept / subject=profnumber; 
run;







The output above gives the results in Table 6.7. Specifically, the fixed-effect estimate for the ‘gpawc’ variable corresponds to the Level 1 GPA effect and the estimate for the ‘gpameans’ variable corresponds to the Level 2 GPA effect.

Alternative approach

Equivalent results can be obtained using the original, un-centered GPA variable along with the Level GPA cluster means as predictors of course evaluation:

	proc mixed data=ch6data; 
 class profnumber;
 model eval = gpa gpameans / solution cl;
 random intercept / subject=profnumber; 
run;







The output above gives the results in Table 6.8.


Revisiting the random-slopes model

Using the cluster-mean centered version of the Level 1 GPA variable leads to random-slopes model which converges easily:

	proc mixed data=ch6data; 
 class profnumber;
 model eval = gpawc / solution;
 random intercept gpawc / subject=profnumber; 
run;





The output above corresponds to the results in Table 6.10 and the random effects reported in the text. 


Formal model comparisons

Here, the goal is to compare a random-intercepts model for the regression of course evaluation on student GPA with a random-slopes model.

But we need to make sure that the same student GPA variable is used in both models: Above, the random-slopes model was estimated using the cluster-mean centered GPA variable. So we need to re-estimate the random intercepts model using the cluster-mean centered GPA variable:

	proc mixed data=ch6data covtest; 
 class profnumber;
 model eval = gpawc / solution;
 random intercept / subject=profnumber; 
run;





Below, we use the difference between the “-2 Res Log Likelihood” values from the outputs to calculate a chi-square test to compare the random-intercepts model for the GPA effect with the random-slopes model for the GPA effect:

	data temp;
 chistat = 605.8 - 605.5;
 cdf = 6 - 4;
 pvalue = 1 - probchi(chistat, cdf);
run;

proc print data=temp;
run;





In the output above, the ‘chistat’ value corresponds to the value of the chi-square statistic reported in Chapter 6 (with some rounding error). The degrees of freedom for this chi-square test can be discerned by subtracting the df value of Model 2 (df=6) from the df value of Model 1 (df=4), leading to the chi-square distribution used to obtain the reported p-value (0.86071). 


Cross-level interactions

Specify and estimate random-slopes model including Level 1 predictor (student GPA), Level 2 predictor (instructor attractiveness), and their interaction.

As explained in Chapter 6, it is important for this model also to include the means of the Level 1 variable (GPA means) as a Level 2 predictor (along with its interaction with the other Level 2 predictor, instructor attractiveness).

First, product variables (bgpa_int and bgpam_int) are created in a data step and then these are used in the ‘mixed’ procedure to represent the interactions:

	data ch6data;
 set ch6data;
 bgpa_int = beauty*gpa;
 bgpam_int = beauty*gpameans;
run;

proc mixed data=ch6data covtest; 
 class profnumber;
 model eval = gpa beauty gpameans bgpa_int bgpam_int / solution;
 random intercept gpa / subject=profnumber; 
run;







The output above corresponds to the results presented in Table 6.11.


Assumption Checking for MLM

As presented in Chapter 6, return to the random-intercepts model with separate within- and between-clusters effect for student GPA.

To reproduce Figure 6.7, we first extract the estimated cluster-specific coefficients from this model:

	proc mixed data=ch6data;
 class profnumber;
 model eval = gpawc gpameans / ddfm=kr solution residual outp=resid;
 random intercept / type=un subject=profnumber solution;
 ods output solutionf=sf(keep=effect estimate
                         rename=(estimate=overall));
 ods output solutionr=sr(keep=effect profnumber estimate
                         rename=(estimate=ssdev));
run;

proc print data=sf;  
run;



data temp;
 set sr;
 Intercept = 3.1247 + ssdev;
 drop Effect ssdev;
run; 

data level2;
 merge temp ch6j94;
 by profnumber;
run;



The code above creates a new dataset (called ‘level2’) with one row for each cluster (i.e., each instructor) and the first three columns corresponding to the estimated coefficients from the model. Take a look at the first six rows of this new dataset, which correspond to the first six instructors (i.e., clusters):

	proc print data=level2 (obs=6);
run;






Notice that each instructor has her own value for the intercept. The remaining columns give the within-cluster (instructor-level) means of the variables that were Level 1 (student-level) variables, which are ‘eval’ and ‘gpa’, and the values of the Level 2 (instructor-level) variables.

Now we can use the intercept values from this Level 2 (instructor-level) data frame to plot these random intercepts against the instructor attractiveness variable, as shown in Figure 6.7:

	proc sgplot data=level2;
 scatter x=beauty y=intercept; 
 reg x=beauty y=intercept;
 xaxis label="Instructor attractiveness";
 yaxis label="Random intercepts";
run;
[image: ]



Next, reproduce Figure 6.8, showing the distribution of the Level 1 residuals from the random-intercepts model:

	proc sgplot data=resid;
 histogram Resid;
 density Resid/lineattrs=(color=black pattern=2);
 density Resid/type=kernel lineattrs=(color=black); 
run;
[image: ]



Figure 6.9 is based on the model with instructor attrativeness added as a Level 2 predictor of random-intercepts for the within-cluster regressions of course evaluation on student GPA. But the model did not disentangle the within-cluster effect of student GPA from the between-cluster effect, so we need to re-estimate that model:

	proc mixed data=ch6data;
 class profnumber;
 model eval = gpa beauty / ddfm=kr solution residual outp=resid;
 random intercept / type=un subject=profnumber solution;
 ods output solutionf=sf(keep=effect estimate
                         rename=(estimate=overall));
 ods output solutionr=sr(keep=effect profnumber estimate
                         rename=(estimate=ssdev));
run;



Again, need to create new instructor-level data frame including the intercept estimates from ch6mod5r:

	proc print data=sf;  
run;


data temp;
 set sr;
 Intercept = 3.5484 + ssdev;
 drop Effect ssdev;
run; 



Next, calculate intercept residuals (difference between each instructor's mean course evaluation and her intercept, or predicted mean, from the random-intercepts model):

	proc means data=ch6data noprint nway ;
 class profnumber;
 var eval;
 output out=evalJ mean=eval_mean;
run;

data evalJ;
 merge evalJ temp;
 by profnumber;
 drop _TYPE_ _FREQ_;
run;

data intresid;
 set evalJ;
 intresid = eval_mean - Intercept;
run;
 
data intresid;
 merge intresid ch6j94;
 by profnumber;
run;



And finally plot the Level 2 residuals against the Level 2 predictor to reproduce Figure 6.9:

	proc sgplot data=intresid;
 scatter x=beauty y=intresid; 
 xaxis label="Instructor attractiveness";
 yaxis label="Random intercept residuals";
run;
[image: ]

